

```

/*
 * Recupera uno stream di dati via HTTP.
 *
 *     i_url --> risorsa HTTP.
 *     i_o_clob --> oggetto recuperato.
 */
PROCEDURE Get_Stream(
    i_url IN VARCHAR2,
    i_o_clob IN OUT CLOB
)
IS

    v_http_request UTL_HTTP.REQ;
    v_http_response UTL_HTTP.RESP;
    v_buffer VARCHAR2(32767);
    v_stato NUMBER;
    v_size NUMBER;

BEGIN

    v_stato := 0;

    -- Connessione HTTP
    v_http_request := UTL_HTTP.Begin_Request(url => i_url, method => 'GET',
    http_version=> 'HTTP/1.0');
    UTL_HTTP.set_header(v_http_request, 'Content-Type', 'text/xml; charset=utf-8');
    v_http_response := UTL_HTTP.Get_Response(v_http_request);

    v_stato := 1; -- Stato 1: CLOB non ancora creato, ma connessione aperta

    -- Lettura dello stream di dati
    BEGIN
        LOOP
            UTL_HTTP.read_text(v_http_response, v_buffer, 32767);
            DBMS_LOB.Writeappend(i_o_clob, LENGTH(v_buffer), v_buffer);
        END LOOP;
    EXCEPTION
        WHEN UTL_HTTP.END_OF_BODY THEN
            UTL_HTTP.End_Response(v_http_response);
    END;

    v_stato := 2; -- Stato 2: CLOB creato e connessione chiusa

```

```
EXCEPTION
  WHEN OTHERS THEN

    IF v_stato = 1 THEN
      UTL_HTTP.End_Response(v_http_response);
    END IF;
    RAISE;

END Get_Stream;
```